
How to use :id in filepath
    

In order to use :id attribute within file's path, we should separate
creation stage on two steps:
	persist the resource
	upload the file

It needs to be done because :id not yet exist on creation stage.
Example
To implement this we should define storage_dir/2 inside uploader
def storage_dir(_version, {_file, scope}) do
  "uploads/avatar/#{scope.id}"
end
Then define two separate changesets inside our resource
def changeset(user, attrs) do
  user
  |> cast(attrs, [:name])
  |> validate_required([:name])
end

def avatar_changeset(user, attrs) do
  user
  |> cast_attachments(attrs, [:avatar])
  |> validate_required([:avatar])
end
Finally, we can combine two stages into one action
Ecto.Multi.new()
|> Ecto.Multi.insert(:user, User.changeset(user, attrs))
|> Ecto.Multi.update(:user_with_avatar, &User.avatar_changeset(&1.user, attrs))
|> Repo.transaction()
This can be used in a Phoenix app context like
defmodule Accounts do
  # ...
  
  def create_user(attrs \\ %{}) do
    Ecto.Multi.new()
    |> Ecto.Multi.insert(:user, User.changeset(%User{}, attrs))
    |> Ecto.Multi.update(:user_with_avatar, &User.avatar_changeset(&1.user, attrs))
    |> Repo.transaction()
    |> case do
      {:ok, %{user_with_avatar: user}} -> {:ok, user}
      {:error, _, changeset, _} -> {:error, changeset}
  end
  
  def update_user(%User{} = user, attrs) do
    user
    |> User.changeset(attrs)
    |> User.avatar_changeset(attrs)
    |> Repo.update()
  end
  
  def change_user(%User{} = user, attrs \\ %{}) do
    user
    |> User.changeset(attrs)
    |> User.avatar_changeset(attrs)
  end
  
  # ...
end



  

    
Waffle.Ecto 
    



      
Waffle.Ecto provides an integration with Waffle and Ecto.
Waffle attachments should be stored in a string column, with a name indicative of the attachment.
create table :users do
  add :avatar, :string
end
How to add waffle_ecto to your project
	add Schema to the model
	configure Definition for uploader

          @type t() :: %{file_name: String.t(), updated_at: DateTime.t()}


      



  


        

      



  

    
Waffle.Ecto.Definition 
    



      
Provides a set of functions to ease integration with Waffle and Ecto.
In particular:
	Definition of a custom Ecto Type responsible for storing the images
	URL generation with a cache-busting timestamp query parameter

Example
defmodule MyApp.Uploaders.AvatarUploader do
  use Waffle.Definition
  use Waffle.Ecto.Definition

  # ...
end
URL generation
Both public and signed urls will include the timestamp for cache
busting, and are retrieved the exact same way as using Waffle
directly.
user = Repo.get(User, 1)

# To receive a single rendition:
MyApp.Uploaders.AvatarUploader.url({user.avatar, user}, :thumb)
#=> "https://bucket.s3.amazonaws.com/uploads/avatars/1/thumb.png?v=63601457477"

# To receive all renditions:
MyApp.Uploaders.AvatarUploader.urls({user.avatar, user})
#=> %{original: "https://.../original.png?v=1234", thumb: "https://.../thumb.png?v=1234"}

# To receive a signed url:
MyApp.Uploaders.AvatarUploader.url({user.avatar, user}, signed: true)
MyApp.Uploaders.AvatarUploader.url({user.avatar, user}, :thumb, signed: true)

      





  

    
Waffle.Ecto.Schema 
    



      
Defines helpers to work with changeset.
Add a using statement use Waffle.Ecto.Schema to the top of your
ecto schema, and specify the type of the column in your schema as
MyApp.Avatar.Type.
Attachments can subsequently be passed to Waffle's storage though a
Changeset cast_attachments/3 function, following the syntax of
cast/3.
Example
defmodule MyApp.User do
  use MyApp.Web, :model
  use Waffle.Ecto.Schema

  schema "users" do
    field :name,   :string
    field :avatar, MyApp.Uploaders.AvatarUploader.Type
  end

  def changeset(user, params \\ :invalid) do
    user
    |> cast(params, [:name])
    |> cast_attachments(params, [:avatar])
    |> validate_required([:name, :avatar])
  end
end

        cast_attachments(changeset_or_data, params, allowed, options \\ [])

      


        Extracts attachments from params and converts it to the accepted format.

Extracts attachments from params and converts it to the accepted format.

  
  options

  
  Options


	:allow_urls — fetches remote file if the string matches ~r/^https?:\/\//
	:allow_paths — accepts any local path as file destination


  
  examples

  
  Examples


cast_attachments(changeset, params, [:fetched_remote_file], allow_urls: true)

Waffle.Ecto.Type 
    



      
Provides implementation for custom Ecto.Type behaviour.

